Ricerca Operativa

Laboratorio: utilizzo di solver per programmazione matematica

AMPL

- **A M**athematicl **P**rogramming **L**anguage
- Linguaggio di modellazione algebrica
  - Esprime un problema di ottimizzazione in una forma comprensibile ad un solutore
  - Linguaggio algebrico: contiene diverse primitive per esprimere la notazione matematica normalmente utilizzata per problemi di ottimizzazione (es. sommatorie, funzioni matematiche, etc.)
- Caratteristiche:
  - Flessibilità: separazione modello / dati, script
  - Integrazione con motori di ottimizzazione allo stato dell'arte
  - Linguaggio «semplice»: traduzione del modello matematico
  - Nuovo linguaggio

Esempio base: contadino

```
#DICHIARAZIONE VARIABILI

var xL >= 0;        #ettari a lattuga
var xP >= 0;        #ettari a patate


#MODELLO

maximize    Resa:       3000 * xL + 5000 * xP;

subject to  ettari:     xL + xP <= 12;
subject to  semi:       7 * xL <= 70;
s.t.        tuberi:     3 * xP <= 18;
s.t.        conc:       10 * xL + 20 * xP <= 160;
```


AMPL: comandi base

```
reset; # cancella dati memorizzati

model contadino.mod; # carica il modello
data contadino.dat; # carica i dati

option solver cplex; # scelta del motore di
                       ottimizzazione
solve; # risolve il modello

display xL, xP; # visualizza il valore
                  (ottimo) delle variabili
```

- **Script:** scrivere i comandi in un file « .run » e
  `include contadino.run;`

Esempio base: contadino [risorse]

- **`contadino.mod`**
- **`contadino.dat`**
- **`contadino.run`**

Esercizio 2.

Risolvere il problema dei telecomandi con AMPL

Per le variabili intere:

```
var xA >=0 integer;
```

[Risorse]

- **`telecomandi.mod`**
- **`telecomandi.dat`**
- **`telecomandi.run`**

Esempio: sintassi comando **var**

- Dichiarazione di variabili:

```
var nomeVar  [>= LB] [, <= UB]
                    [integer/binary];
```

- Esercizio: … al massimo 5 telecomandi di tipo A.

```
var xA >=0 , <= 5 integer;
```


Modelli generali

- I modelli precedenti includono i «dati» del problema:
  - Se cambiano i dati bisogna cambiare il modello
  - Poca leggibilità
  - Difficile riportare modifiche del modello
- Separare modello e dati
  - File « .mod » con il modello «generale» e la definizione dei parametri del problema
  - File « .dat » con i dati attribuiti ai parametri
- Per uno stesso modello possiamo utilizzare diversi file dati (ad esempio «contadino» e «telecomandi»)

Modello «generale»

Modelli di mix ottimo di produzione

$$\max \sum_{i \in I} P_i x_i$$
$$s.t.$$
$$\sum_{i \in I} A_{ij} x_i \le Q_j \qquad \forall\, j \in J$$
$$x_i \in \mathbb{R}_+ \; [\, \mathbb{Z}_+ \mid \{0,1\} \,] \quad \forall\, i \in I$$

I insiene dei beni che possono essere prodotti;

J insieme delle risorse disponibili;

P_i profitto (unitario) per il bene $i \in I$;

Q_j quantità disponibile della risorsa $j \in J$;

A_{ij} quantità di risorsa j necessaria per la produzione di un'unità del bene i.

Modello in AMPL: sintatti base

Espressioni indicizzanti

```
#DICHIARAZIONE INSIEMI
set Prodotti;
set Risorse;

#DICHIARAZIONE PARAMETRI
param maxNumProd;      # massimo numero prodotti
param P {Prodotti};    # profitto unitario
param Q {Risorse};     # disponibilità risorsa
param A {Prodotti,Risorse};
                       # risorsa per unità di pr.

var x {Prodotti} >=0 , <= maxNumProd;

maximize profitto: sum {i in Prodotti} P[i]*x[i];

subject to disponib {j in Risorse}:
         sum {i in Prodotti} A[i,j]*x[i] <= Q[j];
```

Dati in AMPL: sintatti base

```
set Prodotti := lattuga patata;
set Risorse := ettari semi tuberi concime;

param maxNumProd := 7;

param         P  :=
lattuga       3000
patata        5000
;
param         Q  :=
ettari         12
semi           70
tuberi         18
concime       160
;
param A :  ettari semi tuberi concime :=
lattuga    1      7     0      10
patata     1      0     3      20
;
```

A{Prodotti=righe , Risorse=colonne}

Esercizio 3.

Risolvere il problema dei telecomandi con AMPL, usando il modello generale.

- … ogni prodotto ha uno specifico limite superiore!
- .mod

```
param maxNumProd {Prodotti};
var x {i in Prodotti} >=0,<= maxNumProd[i];
```

- .dat

```
param MaxNumProd := telA 5 telB 999;
```

- [risorse]
  - mixOpt.mod - mixOpt.run
  - mixOpt.contadino.dat - mixOpt.telecomandi.dat
- Comando **expand** (visualizza modello esteso)

Definizione e dichiarazione di set e param

- Dichiarazione di insiemi

```
set nome;
```

- Dichiarazione di parametri

```
param nome [{index_expr}] [default value];
```

```
[i in] Set1, [j in] Set2, …
```

- Definizione generale di set e param

```
set nome := elem1 elem2 … elemN;
param nome := index1 index2 … indexN value;
```

- Riferimenti tramite indici (uso di […]):

```
NomeSet[i] #i-esimo elemento di NomeSet
NomeParam[elem1,elem2, …,elemN]
```


Definizione sintetica di parametri

- Dichiarazioni sintetiche: più parametri con stessi indici

```
param : P   MaxNumProd := #notare i due punti dopo "param"
  telA  3   5
  telB  6   999;
```

- Dichiarazioni sintetiche: tabelle

```
param A :   ettari semi tuberi concime := #due punti dopo A
lattuga     1      7    0      10
patata      1      0    3      20;
```

- Dichiarazioni sintetiche: tabelle trasposte

```
param A (tr) :  lattuga patata :=
ettari          1       1
semi            7       0
tuberi          0       3
Concime         10      20;
```

AMPL: installazione

- AMPL è disponibile su http://ampl.com
  - Software commerciale
  - Full per scopi didattici (course edition)
  - Demo version gratuita con max 500 variabili e 300 vincoli
  - http://ampl.com/try-ampl/
- Documentazione
  - Manuale sintetico sulla pagina del corso
  - The AMPL book: http://ampl.com/resources/the-ampl-book/chapter-downloads/

Esercizio 4. Dieta economica

Un dietologo deve preparare una dieta che garantisca un apporto giornaliero di proteine, ferro e calcio di almeno 20 mg, 30 mg e 10 mg, rispettivamente. Il dietologo è orientato su cibi a base di verdura (5 mg/kg di proteine, 6 mg/Kg di ferro e 5 mg/Kg di calcio, al costo di 4 €/Kg), carne (15 mg/kg di proteine, 10 mg/Kg di ferro e 3 mg/Kg di calcio, al costo di 10 €/Kg) e frutta (4 mg/kg di proteine, 5 mg/Kg di ferro e 12 mg/Kg di calcio, al costo di 7 €/Kg). Determinare la dieta di costo minimo.

Risolvere il problema con AMPL (file .mod e .dat separati)

Modello generale per la dieta

Modelli di copertura di costo minimo

$$\begin{aligned} \min \quad & \sum_{i \in I} C_i x_i \\ s.t. \quad & \\ & \sum_{i \in I} A_{ij} x_i \geq D_j & \forall\, j \in J \\ & x_i \in \mathbb{R}_+ \; [\, \mathbb{Z}_+ \mid \{0,1\} \,] & \forall\, i \in I \end{aligned}$$

I insiene delle risorse disponibili;
J insieme delle domande da coprire;
C_i costo (unitario) per l'utilizzo della risorsa $i \in I$;
D_j ammontare della domanda di $j \in J$;
A_{ij} capacità (unitaria) della risorsa i di soddisfare la domanda j.

Modello PL

- Siano x_1, x_2 e x_3 le quantità di cibi a base di verdura, carne e frutta, rispettivamente

$$\begin{aligned} &\min \quad 4\,x_1+10\,x_2+7\,x_3 && \text{(costo giornaliero dieta)} \\ &s.t. \\ &\quad 5x_1+15x_2+4x_3 \geq 20 && \text{(proteine)} \\ &\quad 6x_1+10x_2+5x_3 \geq 30 && \text{(ferro)} \\ &\quad 5x_1+3x_2+12x_3 \geq 10 && \text{(calcio)} \end{aligned}$$

$x_i \in \mathbb{R}_+, \; \forall i \in \{1, 2, 3\}$

Esercizio 5.

Il dietologo vuole inserire alimenti a base di pesce azzurro (10 mg/kg di proteine, 15 mg/kg di ferro e 2 mg/kg di calcio, al costo di 3 euro/kg) nella dieta.

Modificare opportunamente i file relativi al problema.

- [Risorse]
  - `diet.mod – diet.run – diet.1.dat – diet.2.dat`
- Comandi
  - `include diet.run;`
  - `reset data; data diet.2.dat;`
  - `solve; display x;`

Esercizio 6. Indagine di mercato

Un'azienda pubblicitaria deve svolgere un'indagine di mercato per lanciare un nuovo prodotto. Si deve contattare telefonicamente un campione significativo di persone: almeno 150 donne sposate, almeno 110 donne non sposate, almeno 120 uomini sposati e almeno 100 uomini non sposati. Le telefonate possono essere effettuate al mattino (al costo operativo di 1.1 euro) o alla sera (al costo di 1.6 euro). Le percentuali di persone mediamente raggiunte sono riportate in tabella.

	Mattino	Sera
Donne sposate	30%	30%
Donne non sposate	10%	20%
Uomini sposati	10%	30%
Uomini non sposati	10%	15%
Nessuno	40%	5%

Si noti come le telefonate serali sono più costose, ma permettono di raggiungere un maggior numero di persone: solo il 5% va a vuoto. Si vuole minimizzare il costo complessivo delle telefonate da effettuare (mattina/sera) in modo da raggiungere un campione significativo di persone

Risolvere il problema con AMPL (usare soluzione Es. 4)

Modello PLI

- Siano x_1 e x_2 il numero di telefonate da fare al mattino e alla sera, rispettivamente

$$
\begin{array}{lll}
\min & 1.1\,x_1+1.6\,x_2 & \text{(costo totale telefonate)} \\
s.t. & & \\
& 0.3x_1+0.3x_2 \geq 150 & \text{(donne sposate)} \\
& 0.1x_1+0.2x_2 \geq 110 & \text{(donne non sposate)} \\
& 0.1x_1+0.3x_2 \geq 120 & \text{(uomini sposati)} \\
& 0.1x_1+0.15x_2 \geq 100 & \text{(uomini non sposati)}
\end{array}
$$

$x_i \in \mathbb{Z}_+, \ \forall i \in \{1, 2\}$

[Risorse] `diet.indagine.dat`

Esercizio 7. Trasporto di frigoriferi

Una ditta di produzione di elettrodomestici produce dei frigoriferi in tre stabilimenti e li smista in quattro magazzini intermedi di vendita. La produzione settimanale nei tre stabilimenti A, B e C è rispettivamente di 50, 70 e 20 unità. La quantità richiesta dai 4 magazzini è rispettivamente di 10, 60, 30 e 40 unità. I costi per il trasporto di un frigorifero tra gli stabilimenti e i magazzini 1, 2, 3 e 4 sono i seguenti:

- dallo stabilimento A: 6, 8, 3, 4 euro;
- dallo stabilimento B: 2, 3, 1, 3 euro;
- dallo stabilimento C: 2, 4, 6, 5 euro.

Utilizzare AMPL per determinare il piano di trasporti di costo minimo, considerando che non sono ammesse rimanenze alla fine della settimana e che lo stesso modello dovrà essere utilizzato per diverse settimane.

Modello PLI

- Sia x_{ij} il numero di frigoriferi prodotti nello stabilimento i e smistati nel magazzino j

$$
\begin{array}{lll}
\min & 6\,x_{A1}+8\,x_{A2}+3\,x_{A3}+4\,x_{A4}+ & \\
+ & 2\,x_{B1}+3\,x_{B2}+1\,x_{B3}+3\,x_{B4}+ & \\
+ & 2\,x_{C1}+4\,x_{C2}+6\,x_{C3}+5\,x_{C4} & \\
s.t. & & \\
& x_{A1}+x_{A2}+x_{A3}+x_{A4} \le 50 & \text{(capacità produttiva stabilimento A)} \\
& x_{B1}+x_{B2}+x_{B3}+x_{B4} \le 70 & \text{(capacità produttiva stabilimento B)} \\
& x_{C1}+x_{C2}+x_{C3}+x_{C4} \le 20 & \text{(capacità produttiva stabilimento C)} \\
& x_{A1}+x_{B1}+x_{C1} \ge 10 & \text{(domanda magazzino 1)} \\
& x_{A2}+x_{B2}+x_{C2} \ge 60 & \text{(domanda magazzino 2)} \\
& x_{A3}+x_{B3}+x_{C3} \ge 30 & \text{(domanda magazzino 3)} \\
& x_{A4}+x_{B4}+x_{C4} \ge 40 & \text{(domanda magazzino 4)}
\end{array}
$$

$$x_{ij} \in \mathbb{Z}_+ \ \forall i \in \{A, B, C\}, j \in \{1, 2, 3, 4\}$$

Modello «generale» per trasporti

Modelli di trasporto

$$
\begin{array}{lll}
\max & \sum_{i \in I} \sum_{j \in J} C_{ij} x_{ij} & \\
s.t. & & \\
& \sum_{j \in J} x_{ij} \le O_i & \forall\, i \in I \\
& \sum_{i \in I} x_{ij} \ge D_j & \forall\, j \in J \\
& x_{ij} \in \mathbb{R}_+ \ [\, \mathbb{Z}_+ \mid \{0,1\} \,] & \forall\, i \in I, j \in J
\end{array}
$$

I insiene dei centri di offerta; O_i ammontare dell'offerta in $i \in I$;

J insieme dei centri di domanda; D_j ammontare della domanda in $j \in J$.

C_{ij} costo (unitario) per il trasporto da $i \in I$ a $j \in J$;

`display` e espressioni indicizzanti

- Visualizza elementi del modello e della soluzione

  display *elemento1*[, *elemento2*, …];

  display {*ind_expr*} *elemento*[*indici*]

- Esempi di espr. indicizzanti *ind_expr* (condizionate)
  - `display {i in I} C[i,"m3"];`
  - `display {i in I, j in J: C[i,j] >= 5} C[i,j];`
  - `display {i in I: origine[i].body - O[i] != 0} O[i]-origine[i].body;`

- Le espressioni indicizzanti possono essere usate anche per la definizione di variabili, parametri, vincoli etc.

- [Risorse]: `trasporto.mod` – `trasporto_frigo.dat` – `trasporto_frigo.run`